
Introduction

We have been successfully applying the OTS/CafeOBJ method to

modeling and verification of distributed systems such as security

protocols.

• A system is modeled as an observational transition system, or an OTS.

• The OTS is written in CafeOBJ.

• Properties to be proved are expressed as CafeOBJ terms.

• Proofs, or proof scores showing that the OTS has the properties are

written in CafeOBJ.

• The proof scores are veriified by executing them with the CafeOBJ

system.

Outline of Talk

• Observational Transition Systems (OTSs)

• Example: Queuing Lock

• Compositionally Writing Proof Scores

• Ongoing and Future Work

– Generating Proof Scores

– Model-Checking OTS/CafeOBJ specifications

Observational Transition Systems

Observational Transition Systems

An OTS S consists of 〈O, I, T 〉:

• O : A set of observers.

Each o ∈ O is a function o : Υ→ D, where D is a data type.

υ1 =S υ2
def= ∀o ∈ O.o(υ1) = o(υ2) .

• I : A set of initial states.

• T : A set of conditional transition rules.

Each τ ∈ T is a function τ : Υ/=S → Υ/=S on equivalence classes of

Υ wrt =S .

The condition cτ of a transition rule τ ∈ T is called the effective

condition.

Observational Transition Systems

Executions and Invariants

An execution of S is an infinite seuqence υ0, υ1, . . . of states satisfying:

• Initiation : υ0 ∈ I.

• Consecution : For each i ∈ {0, 1, . . .}, υi+1 =S τ (υi) for some τ ∈ T .

A state is called reachable wrt S iff there exists an execution of S in

which the state appears.

Let RS be the set of all the reachable states wrt an OTS S .

If predicate p is true in every state of RS , p is called invariant to S,

which is defined as follows:

invariant p def= ∀υ ∈ RS. p(υ) .

Observational Transition Systems

Indexed Observers and Transition Rules

Observers and transition rules may be indexed.

• Observers are generally denoted by oi1,...,im.

• Transition rules are generally denoted by τj1,...,jn.

where m,n ≥ 0 and there exists data types Dk such that

k ∈ Dk (k = i1, . . . , im, j1, . . . , jn).

For example,

• ap : Observer denoting an interger array a possessed by a process p.

• inc-ap,i : Transition rule denoting the increment of the ith element of

the array.

Specification of OTSs in CafeOBJ

CafeOBJ

• Two kinds of sorts.

– Visible sort denotes an abstract data type.

– Hidden sort denotes the state space of an abstract machine.

• Two kinds of operators for hidden sorts.

– Action operators denote state transitions of an abstract machine.

– Only observation operators can be used to know the inside of an

abstract machine.

Specification of OTSs in CafeOBJ

Writing OTSs in CafeOBJ

• The state space Υ is denoted by a hidden sort, say H.

• An observer oi1,...,im is denoted by an observation operator.

bop o : H Vi1 . . . Vim -> V

• A transition rule τj1,...,jn is denoted by an action operator.

bop a : H Vj1 . . . Vjn -> H

• τj1,...,jn is defined with equations by describing how the value returned

by each observer oi1,...,im changes.

eq o(a(S,Xj1, . . . ,Xjn),Xi1, . . . ,Xim) = NewValue

if c-a(S,Xj1, . . . ,Xjn) .

c-a denotes the effective condition of τj1,...,jn.

Example: Queuing Lock

Queuing Lock

Program executed by process i:

l1: put(queue, i)

l2: repeat until top(queue) = i

Critical Section

cs: get(queue)

• queue is the queue of process IDs shared by all processes.

• put(queue, i) puts i into queue at the end.

• top(queue) returns the top of queue.

• get(queue) removes the top of queue.

Example: Queuing Lock

Modeling Queuing Lock as an OTS

• Observers.

– queue returns the queue shared by all processes. It initially returns

the empty queue.

– pci (i ∈ Pid) returns the label of a command that process i will

execute next. Each pci initially returns label l1.

• Transition rules.

– wanti (i ∈ Pid) denotes that process i executes the command at

label l1.

– tryi (i ∈ Pid) denotes that process i executes one iteration of the

loop at label l2.

– exiti (i ∈ Pid) denotes that process i executes the command at

label cs.

Example: Queuing Lock

Description of the OTS in CafeOBJ (1)

Signature of the CafeOBJ specification of the OTS:

*[Sys]*

-- any initial state

op init : -> Sys

-- observations

bop pc : Sys Pid -> Label

bop queue : Sys -> Queue

-- actions

bop want : Sys Pid -> Sys

bop try : Sys Pid -> Sys

bop exit : Sys Pid -> Sys

• Observers.

– queue

– pci (i ∈ Pid)

• Transition rules.

– wanti (i ∈ Pid)

– tryi (i ∈ Pid)

– exiti (i ∈ Pid)

Example: Queuing Lock

Description of the OTS in CafeOBJ (2)

Action operator want is defined with the equations:

op c-want : Sys Pid -> Bool

eq c-want(S,I) = (pc(S,I) = l1) .

--

ceq pc(want(S,I),J)

= (if I = J then l2 else pc(S,J) fi) if c-want(S,I) .

ceq queue(want(S,I)) = put(queue(S),I) if c-want(S,I) .

ceq want(S,I) = S if not c-want(S,I) .

Example: Queuing Lock

Description of the OTS in CafeOBJ (3)

Action operator try is defined with the equations:

op c-try : Sys Pid -> Bool

eq c-try(S,I) = (pc(S,I) = l2 and top(queue(S)) = I) .

--

ceq pc(try(S,I),J)

= (if I = J then cs else pc(S,J) fi) if c-try(S,I) .

eq queue(try(S,I)) = queue(S) .

ceq try(S,I) = S if not c-try(S,I) .

Example: Queuing Lock

Description of the OTS in CafeOBJ (4)

Action operator exit is defined with the equations:

op c-exit : Sys Pid -> Bool

eq c-exit(S,I) = (pc(S,I) = cs) .

--

ceq pc(exit(S,I),J)

= (if I = J then l1 else pc(S,J) fi) if c-exit(S,I) .

ceq queue(exit(S,I)) = get(queue(S)) if c-exit(S,I) .

ceq exit(S,I) = S if not c-exit(S,I) .

Example: Queuing Lock

Mutual Exclusion

To prove the queuing lock mutually exclusive, all we have to do is to

prove that inv1 is invariant to the OTS modeling the queuing lock.

eq inv1(S,I,J)

= (pc(S,I) = cs and pc(S,J) = cs implies I = J) .

To this end, we need the following:

eq inv2(S,I) = (pc(S,I) = cs implies top(queue(S)) = I) .

eq inv3(S,I) = (pc(S,I) = l2 or pc(S,I) = cs

implies not empty?(queue(S))) .

eq inv4(S,I) = (pc(S,I) = l2 implies I \in queue(S)) .

Example: Queuing Lock

Strategies of Proofs

• First prove that inv2, inv3 and inv4 are invariant, and then prove

that inv1 is invariant using the proved invariants.

inv1

inv4

inv3

inv2

• Prove that inv1 ∧ inv2 ∧ inv3 ∧ inv4 is invariant.

• Prove that inv1, inv2, inv3 and inv4 are invariant simultaneously.

Example: Queuing Lock

Proof of inv1

• Consider the inductive case where try preserves inv1.

We should prove inv1(s, i, j)⇒ inv1(try(s, k), i, j).

• inv2 is used to strengthen the inductive hypothesis.

(inv2(s, i) ∧ inv2(s, j) ∧ inv1(s, i, j))⇒ inv1(try(s, k), i, j)

•When the effective condition is true, the case is split into 4 subcases.

1. (i = k) ∧ (j = k), which needs nothing.

2. (i = k) ∧ (j 6= k), which needs inv2(s, j).

3. (i 6= k) ∧ (j = k), which needs inv2(s, i).

4. (i 6= k) ∧ (j 6= k), which needs nothing.

Example: Queuing Lock

Proof Passage of inv1

Proof passage of subcase 2 ((i = k) ∧ (j 6= k)):

open ISTEP

-- arbitrary objects

op k : -> Pid .

-- assumptions

-- eq c-try(s,k) = true .

eq pc(s,k) = l2 . eq top(queue(s)) = k .

--

eq i = k . eq (j = k) = false .

-- successor state

eq s’ = try(s,k) .

-- check; istep1(i,j) = inv1(s,i,j) implies inv1(s’,i,j)

red inv2(s,j) implies istep1(i,j) .

close

Example: Queuing Lock

Proof of inv2

• Consider the inductive case where exit preserves inv2.

We should prove inv2(s, i)⇒ inv2(exit(s, k), i).

• inv1 is used to strengthen the inductive hypothesis.

(inv1(s, i, k) ∧ inv2(s, i))⇒ inv2(try(s, k), i)

•When the effective condition is true, the case is split into 3 subcases.

1. i = k, which needs nothing.

2. (i 6= k) ∧ (pc(s, i) = cs), which needs inv1(s, i, k).

3. (i 6= k) ∧ (pc(s, i) 6= cs), which needs nothing.

Example: Queuing Lock

Proof Passage of inv2

Proof passage of subcase 2 ((i 6= k) ∧ (pc(s, i) = cs)):

open ISTEP

-- arbitrary objects

op k : -> Pid .

-- assumptions

-- eq c-exit(s,k) = true .

eq pc(s,k) = cs .

--

eq (i = k) = false . eq pc(s,i) = cs .

-- successor state

eq s’ = exit(s,k) .

-- check; istep2(i) = inv2(s,i) implies inv2(s’,i)

red inv1(s,i,k) implies istep2(i) .

close

Compositionally Writing Proof Scores

Compositional Proofs of Invariants (1)

• Consider proving that pred 1(s, x1) is invariant to an OTS.

• Together with pred 2(s, x2), . . . , predn(s, xn).

Let pred(s, x1, . . . , xn) be pred 1(s, x1) ∧ . . . predn(s, xn).

• Consider the inductive case where an action operator a preserves

pred(s, x1, . . . , xn).

We should prove

pred(s, x1, . . . , xn)⇒ pred(a(s, y), x1, . . . , xn)

or

pred 1(s, x1)⇒ pred 1(a(s, y), x1)
...

predn(s, xn)⇒ predn(a(s, y), xn)

(1)

Compositionally Writing Proof Scores

Compositional Proofs of Invariants (2)

• Let pred i(s, xi)⇒ pred i(a(s, y), xi) be one of such formulas.

•What strengthens the inductive hypothesis can be

pred j1
(s, tj1) ∧ . . . ∧ pred jk

(s, tjk), where 1 ≤ j1, . . . , jk ≤ n ,

where tj is a term denoting an instance of xj.

Let SIH i be pred j1
(s, tj1) ∧ . . . ∧ pred jk

(s, tjk).

• The proof of the ithe formula of (1) can be replaced with the proof of

the formula:

(SIH i ∧ pred i(s, xi))⇒ pred i(a(s, y), xi) (2)

Compositionally Writing Proof Scores

Compositional Proofs of Invariants (3)

(SIH 1 ∧ pred 1(s, x1))⇒ pred 1(a(s, y), x1)
...

(SIH n ∧ predn(s, xn))⇒ predn(a(s, y), xn)

where SIH i = pred ji
1
(s, tji

1
) ∧ . . . ∧ pred ji

ki
(s, tji

ki
), i = 1, . . . , n.

From these n formulas, the following can be deduced.

(SIH 1 ∧ . . . ∧ SIH n) ∧ (pred 1(s, x1) ∧ . . . ∧ predn(s, xn))

⇒ (pred 1(a(s, y), x1) ∧ . . . ∧ predn(a(s, y), xn))

SIH 1 ∧ . . . ∧ SIH n can be used as the inductive hypothesis becasue

x1, . . . , xn are just instantiated.

Therefore, if this formula is proved, then it is shown that action operator

a preserves pred 1(s, x1) ∧ . . . ∧ predn(s, xn).

Compositionally Writing Proof Scores

Compositional Proofs of Invariants (4)

• The case may have to be split into multiple subcases to prove

(SIH i ∧ pred i(s, xi))⇒ pred i(a(s, y), xi) (2)

• Suppose that the case is split into l subcases denoted by

case i
1, . . . , case

i
l, which should satisfy

(case i
1 ∨ . . . ∨ case i

l) = true .

• The proof of (2) can be replaced with the proofs of the l formulas:

(case i
1 ∧ SIH i ∧ pred 1(s, xi))⇒ pred 1(a(s, y), xi)

...

(case i
l ∧ SIH i ∧ pred 1(s, xi))⇒ pred 1(a(s, y), xi)

Compositionally Writing Proof Scores

Proof Scores of Invariants (1)

In module INV:

op inv1 : H V1 -> Bool -- pred 1(s, x1)

. . . -- . . .

op invn : H Vn -> Bool -- predn(s, xn)

eq inv1(S,X1) = pred1(S,X1) .

. . .

eq invn(S,Xn) = predn(S,Xn) .

In module ISTEP:

op istep1 : V1 -> Bool -- pred 1(s, x1)⇒ pred 1(a(s, y), x1)

. . . -- . . .

op istepn : Vn -> Bool -- pred 1(s, xn)⇒ pred 1(a(s, y), xn)

eq istep1(X1) = inv1(s,X1) implies inv1(s
′,X1) .

. . .

eq istepn(Xn) = invn(s,Xn) implies invn(s
′,Xn) .

Compositionally Writing Proof Scores

Proof Scores of Invariants (2)

Proof passage of case i
j where an action operator a preserves predi(s, ti):

open ISTEP

-- arbitary objects

op y : -> V .

Declare constants if necessary

-- assumptions

Declare equations denoting case i
j

-- successor state

eq s′ = a(s, y) .

-- check; istepi(xi) = predi(s, ti) implies predi(s
′, ti)

red (predj1
(s, tj1) and . . . predjk

(s, tjk)) implies istepi(xi) .

close

Compositionally Writing Proof Scores

Some Merits

• Relieve the complexity of the reduction of logical formulas.

Because the compositional writing of proof scores makes it possible to

focus on each conjunct pred i(s, xi) of a large formula

pred 1(s, x1) ∧ . . . predn(s, xn).

• Ease the complexity of case analysis.

Because the compositional writing of proof scores makes it possible to

do case analysis for each conjunct pred i(s, xi) only.

Compositionally Writing Proof Scores

Case Studies

• NSLPK authentication protocol

17 invariants verified; 9,500 lines.

• iKP electronic payment protocols

18 invariants verified; proof scores of 22,000 lines.

• Horn-Preneel micropayment protocol

24 invariants verified; proof scores of 22,000 lines.

• NetBill electronic commerce protocol

36 invariants verified; proof scores of 79,000 lines.

• SET payment protocol

20 invariants verified; proof scores of 40,000 lines.

• TLS handshake protocol

18 invariants verified; proof scores of 14,000 lines.

Ongoing and Future Work

Ongoing and Future Work

• A script language and a tool (called Gateau) that translates scripts

written in the language into proof scores have been being designed.

A prototype has been implemented and used for verifying the queuing

lock and the NSLPK authentication protocol.

• A tool that translates OTS/CafeOBJ specifications into SMV ones has

been designed.

A prototype has been implemented and used for model-checking an

OTS/CafeOBJ specification of the NSPK authentication protocol.

But, it is difficult to translate CafeOBJ user-defined data type into

those encoded in SMV limited data types.

Therefore, we are going to design a tool that translates OTS/CafeOBJ

specifications into Maude ones.

Generating Proof Scores

Gateau Scripts

The Gateau script of inv1 of the queuing lock:

#base: inv1(init,i,j)

#inductive: istep1(i,j)

#successor: s’

#action: want(s,k)
#constants: k : -> Pid
#effective: pc(s,k) = l1
#case: i = k

(i = k) = false
#case: j = k

(j = k) = false

#action: try(s,k)
#constants: k : -> Pid
#effective: pc(s,k) = l2

top(queue(s)) = k
#lemma: inv2(s,i) and inv2(s,j)
#case: i = k

(i = k) = false
#case: j = k

(j = k) = false

#action: exit(s,k)
#constants: k : -> Pid
#effective: pc(s,k) = cs
#case: i = k

(i = k) = false
#case: j = k

(j = k) = false

Generating Proof Scores

Experimental Data of Gateau

• QLOCK – We have verified that the queuing lock is mutually exclusive

with Gateau.

• NSLPK – We have verified that the NSLPK authentication protocol has

the nonce secrecy and one-to-many agreement properties with Gateau.

Examples Size of Gateau Scripts Size of Generated Proof Cores

QLOCK 142 lines 2946 lines

NSLPK 1997 lines 19251 lines

The size of the hand-written proof scores are

QLOCK 966 lines

NSLPK 9664 lines

Model-Checking OTS/CafeOBJ specifications

Writing OTSs in Maude (1)

• O and T are denoted by sorts, say OValue and TRule; Υ is denoted

by a sort, say Sys.

• A snapshot of S is represented by a bag of observers and transition

rules.

subsort OValue TRule < Sys .

op none : -> Sys .

op __ : Sys Sys -> Sys [assoc comm id: none]

Generally, a snapshot of S is as the following form:

ovalue-1 . . . ovalue-M trule-1 . . . trule-N

where ovalue-i (i = 1, . . . ,M) is a term denoting an observer, and

trule-i (i = 1, . . . , N) is a term denoting a transition rule.

The OTS/CafeOBJ Method and Some Future Directions by K. Ogata and K. Futatsugi 31/37



Model-Checking OTS/CafeOBJ specifications

Writing OTSs in Maude (2)

• oi1,...,im : Υ→ D, where k ∈ Dk (k = i1, . . . , im), is denoted by the

operator declared as follows:

op (o[ , . . . , ] : ) : Vi1 . . . Vim V -> OValue .

where Vk (k = i1, . . . , im) and V correspond to Dk and D.

• τj1,...,jn : Υ→ Υ, where k ∈ Dk (k = j1, . . . , jn), is denoted by the

operator declared as follows:

op r : Vj1 . . . Vjn -> TRule .

where Vk (k = j1, . . . , jn) correspond to Dk.

Model-Checking OTS/CafeOBJ specifications

Writing OTSs in Maude (3)

• Transition rules are defined in Maude rules.

Suppose that observers needed and affected by the execution of the

transition rule τj1,...,jn are o
1
i11,...,i

1
m1
, . . . , ol

il1,...,i
l
ml

.

crl [rule-r] :

r(Xj1, . . . ,Xjn)

(o1[Xi11
, . . . ,Xi1m1

] : X1) . . . (o
l[Xil1

, . . . ,Xilml
] : Xl)

=>

r(Xj1, . . . ,Xjn)

(o1[Xi11
, . . . ,Xi1m1

] : X′
1) . . . (o

l[Xil1
, . . . ,Xilml

] : X′
l)

if c-r(Xj1, . . . ,Xjn,Xi11
, . . . ,Xi1m1

,X1, . . . ,Xil1
, . . . ,Xilml

,Xl) .

Model-Checking OTS/CafeOBJ specifications

Writing the Queing Lock in Maude (1)

The operators denoting the observers and transition rules are declared as

follows:

*** Observers

op pc[_] :_ : Pid Label -> OValue .

op queue :_ : Queue -> OValue .

*** Transition rules

op want : Pid -> TRule .

op try : Pid -> TRule .

op exit : Pid -> TRule .

• Observers.

– pci (i ∈ Pid)

– queue

• Transition rules.

– wanti (i ∈ Pid)

– tryi (i ∈ Pid)

– exiti (i ∈ Pid)

Model-Checking OTS/CafeOBJ specifications

Writing the Queuing Lock in Maude (2)

The transition rules are defined with these rules.

crl [want] :

want(P) (pc[P] : L) (queue : Q)

=> want(P) (pc[P] : l2) (queue : put(Q,P))

if L == l1 .

crl [try] :

try(P) (pc[P] : L) (queue : Q)

=> try(P) (pc[P] : cs) (queue : Q)

if L == l2 and top(Q) == P .

crl [exit] :

exit(P) (pc[P] : L) (queue : Q)

=> exit(P) (pc[P] : l1) (queue : get(Q))

if L == cs .

Model-Checking OTS/CafeOBJ specifications

Model-Checking (1)

State predicates needed to describe properties to be checked:

mod QLOCK-PREDS is

pr QLOCK .

inc SATISFACTION .

subsort Sys < State .

op crit : Pid -> Prop .

var P : Pid .

var S : Sys .

eq (pc[P] : cs) S |= crit(P) = true .

endm

Model-Checking OTS/CafeOBJ specifications

Model-Checking (2)

Initial state and LTL formula denoting mutual exclusion:

mod QLOCK-CHECK is

inc QLOCK-PREDS . inc MODEL-CHECKER . inc LTL-SIMPLIFIER .

ops p1 p2 p3 : -> Pid .

op init : -> Sys . op mutex : -> Formula .

eq init = want(p1) try(p1) exit(p1) want(p2) try(p2) exit(p2)

want(p3) try(p3) exit(p3)

(pc[p1] : l1) (pc[p2] : l1) (pc[p3] : l1) (queue : empty) .

eq mutex = ([] ~(crit(p1) /\ crit(p2)))

/\ ([] ~(crit(p1) /\ crit(p3)))

/\ ([] ~(crit(p2) /\ crit(p3))) .

endm

Then, model-check that the finite OTS has the property.

red modelCheck(init,mutex) .

